Tutorial

¿Qué es?

Este documento presenta un breve tutorial por  Brian Doty's Grid Analysis and Display System (GrADS). Las siguientes sesiones ejemplo le darán un sentimiento de como usar las capacidades básicas de GrADS. Esta sesión toma alrededor de 30 minutos para ser ejecutada totalmente.  

Antes de empezar: Descargue el dato de muestra 

Usted necesitara example.tar.gz los siguientes archivos de datos muestra para el desarrollo de este tutorial. After downloading the file, unpack it with the command:
    
    tar xvfz example.tar.gz 

and you should see the following three files: 
model.ctl   (A GrADS descriptor file)

model.dat   (A GrADS binary data file)

tutorial    (A text file containing the contents of this tutorial)
Este archivo de datos es descrito por el archivo descriptor de datos model.ctl. Usted debe dar un vistazo a este archivo antes de continuar. El archivo descriptor de datos describe el actual archive de datos, el cual en el caso contiene 5 días de grillas globales que son 72 x 46 elementos en tamaño. 

Por favor descargue estos 3 archivos al directorio local antes de iniciar los procedimientos.  

Sesión Muestra 

Para iniciar GrADS, digite: 

gradsc 

Si el ejecutable gradsc no esta en su directorio actual, o si no esta en su PATH, usted necesitara digitar la ruta completa, ie: 

/usr/homes/smith/grads 

GrADS presentara un  prompt con una pregunta landscape vs. portrait; sólo presione enter. En este punto una ventana grafica debía aparecer en su consola. Usted puede desear mover o redimensionar esta ventana. Tenga en mente que usted digitará comandos de GrADS de la ventana donde usted primero empezó con GrADS -- esta ventana necesitara ser hecha la ventana ‘activa’ y usted no querrá enteramente cubrir esta ventana con la ventana de salida de gráficos.  
En la ventana de texto (de donde usted inicio grads), usted debe ahora ver un prompt: ga->  Usted digitara comandos de GrADS en este prompt y verá los resultados mostrados en la ventana de salida de gráficos.  

El primer comando que usted digitara es: 

open model.ctl 

Usted puede ver que existe en este archivo  digitando: 

query file 

Una de las variables existentes es la llamada ps, para la presión en superficie. Podemos mostrar esta variable digitando: 

d ps 

d es la forma corta para display. Usted notara que por defecto, GrADS mostrara un grafico en X, Y para el primer tiempo en el nivel inferior para el grupo de datos.  

[image: image1.png]




Ahora usted digitara los comandos para alterar las  dimension environment (dimensiones de entorno). El comando display (and implicitly, the access, operation, and output of the data) hará cambios con respecto a las dimensiones de entorno actuales. Usted control alas dimensiones de entorno con el comando set: 

clear           limpia la pantalla
set lon -90     define longitud a 90 grados Oeste 
set lat 40      define latitud a 40 grados Norte 
set lev 500     define niveles a 500 mb
set t 1         define tiempo al primer tiempo paso 
d hgt             grafica la  variable 'hgt'

En la secuencia de comandos arriba, hemos definido las cuatro dimensiones de GrADS para un valor simple. Cuando definimos una dimensión para un valor simple, decimos que la dimensión es “fija”. Puesto que todas las dimensiones son fijas, cuando graficamos una variable obtenemos un valor simple, en este caso el valor en la localización 90W, 40N, 500mb, y el primer tiempo en el grupo de datos. 

Si ahora se digita: 

set lon -180 0     X es ahora la variación dimensional
d hgt
Se ha definido la dimensión X, o longitud, para variar. Se ha hecho esto digitando dos valores en el comando set. Ahora tenemos una variación dimensional (las otras dimensiones son aun fijas), y cuando mostramos una variable obtenemos una grafica lineal, en este caso un grafico de altura geopotencial a 500mb en 40N. 

Ahora digite: 

clear
set lat 0 90
d hgt
Ahora se tiene dos dimensiones variando, por lo tanto por defecto se obtiene una grafica de contorno. Si se tiene 3 dimensiones variando:
c
set t 1 5
d hgt
Se obtiene una secuencia de animación, en este caso a través del tiempo. 

Ahora digite: 

clear
set lon -90
set lat -90 90
set lev 1000 100
set t 1
d tair
d u
En este caso se ha definido las dimensiones Y (latitud) y Z (nivel) para variar, por lo que se obtiene una sección vertical cruzada. También se ha mostrado dos variables, las cuales simplemente se han superpuesto unas a otras. Usted puede mostrar tantos ítems como usted decida superponiendo antes que digite el comando clear . 

Otro ejemplo, en este caso con X y T variando (grafico Hovmoller): 

c
set lon -180 0
set lat 40
set lev 500
set t 1 5


 HYPERLINK "http://grads.iges.org/grads/gadoc/gradcomddisplay.html" d hgt

Ahora que usted sabe como seleccionar la porción de los datos para visualizar, nos moveremos a los tópicos de operaciones con los datos. Primero, defina las dimensiones de entorno a un Z, Y variando uno: 

clear
set lon -180 0
set lat 0 90
set lev 500 
set t 1
Ahora supongamos que queremos ver la temperatura en Fahrenheit en lugar de Kelvin. Podemos hacer la conversión digitando: 

display (t-273.16)*9/5+32 

Alguna expresión puede ser digitada para que invoque las operadores  estándar de +,-,*, y /, las cuales invocan operaciones las cuales pueden ser constante, variables, o funciones. Un ejemplo de invocación de funciones:

clear
d sqrt(u*u+v*v)
Para calcular la magnitud de viento. Una función es proporcionada para hacer este cálculo directamente:  

d mag(u,v) 

Otra construcción en funciones es la función promedio: 

clear
d ave(hgt,t=1,t=5)
En este caso calculamos el promedio de 5 días. Podemos también remover el promedio del campo actual: 

d hgt - ave(hgt,t=1,t=5) 

Podemos también tomar promedios sobre longitud para remover la media zonal: 

clear 
d hgt-ave(hgt,x=1,x=72)
d hgt
Podemos también evaluar diferencia en el tiempo: 

clear
d hgt(t=2)-hgt(t=1)

Esto calcula el cambio entre dos campos sobre un día. Podemos hacer este mismo cálculo usando un offset del tiempo actual: 

d hgt(t+1) - hgt 

La especificación de un nombre de variable es: 

name.file(dim +|-|= value, ...) 

Si tenemos dos archivos abiertos, tal vez uno con salidas de modelo, el otro con análisis, podemos tomar la diferencia entre los dos campos  digitando: 

display hgt.2 - hgt.1 

Otra construcción en funciones calcula la vorticidad horizontal relativa vía diferencias finitas: 

clear
d hcurl(u,v)
Y otra función toma una integración vertical de masa pesada (mass weighted): 

clear
d vint(ps,q,275)
Aquí hemos calculado el agua precipitable en mm. 

Ahora nos moveremos hacia el tópico de control de salidas graficas. Particularmente, hemos permitido a GrADS elegir un contorno de intervalo por defecto. Podemos ignorar esto por: 

clear
set cint 30
d hgt
Podemos también controlar el contorno  de colores por: 

clear


 HYPERLINK "http://grads.iges.org/grads/gadoc/gradcomdsetccolor.html" set ccolor 3
d hgt
Podemos seleccionar formas alternas de presentar los datos: 

clear
set gxout shaded
d hcurl(u,v)
Esto no es muy suave; podemos aplicar un suavizador cúbico digitando:  

clear
set csmooth on
d hcurl(u,v)

Podemos superponer diferentes tipos de gráficos: 

set gxout contour
set ccolor 0
set cint 30
d hgt
Y anotar: 

draw title 500mb Heights and Vorticity 

Podemos visualizar  vectores de viento: 

clear
set gxout vector
d u;v

Aquí estamos mostrando dos expresiones, la primera para la componente U del vector; la segunda componente V del vector. Podemos también colorear los vectores por especificación de un tercer campo: 

d u;v;q 

o tal vez: 

d u;v;hcurl(u,v) 

Usted puede graficar seudo vectores, graficando cualquier campo que usted quiera: 

clear
d mag(u,v) ; q*10000
Aquí la componente U es la velocidad de viento y V es la humedad . 

Podemos también ver líneas de corriente (y colorearlos): 

clear
set gxout stream
d u;v;hcurl(u,v)
O podemos mostrar los valores puntuales de grilla:  

clear
set gxout grid
d u

Podemos alterar el mapa de fondo: 

clear
set lon -110 -70
set lat 30 45
set mpdset mres
set digsize 0.2
set dignum 2
d u
Para alterar la proyección: 

set lon -150 -40
set lat 15 80 
set mpvals -120 -75 25 65
set mproj nps
set gxout contour
set cint 30
d hgt
En este caso, le hemos dicho a GrADS que acceda y opere sobre los datos de longitud 140W a 40W, y latitud 15N a 80N. Pero le dijimos que  muestre un  grafico polar estereográfico que contenga la región limitada por 120W a 75W y 25N a 65N. El área extra del grafico es superpuesta por el mapa de la rutina de proyección. 

Esto concluye la sección ejemplo. En este punto, usted  deseara examinar los  datos mas, o usted podría querer ir a través de la documentación de GrADS e intentar otras opciones descritas allí. 

